1620 GENERAL PROGRAM LIBRARY

Computation of Bessel Functions
of Integral Order

3.0.005

DISCLAIMER

Although each program has been tested by its contributor, no warranty, express or implied, is made by the contributor or 1620 USERS Group, as to the accuracy and functioning of the program and related program material, nor shall the fact of distribution constitute any such warranty, and no responsibility is assumed by the contributor or 1620 USERS Group, in connection therewith.

1620 USERS GROUP PROGRAM REVIEW AND EVALUATION

(fill out in typewriter or pencil, do not use ink)

Program No.________________ Date________________

Program Name:____

1. Does the abstract adequately describe what the program is and what it does? Yes___ No___
   Comment____

2. Does the program do what the abstract says? Yes___ No___
   Comment____

3. Is the Description clear, understandable, and adequate? Yes___ No___
   Comment____

4. Are the Operating Instructions understandable and in sufficient detail? Yes___ No___
   Comment____
   Are the Sense Switch options adequately described (if applicable)? Yes___ No___
   Are the mnemonic labels identified or sufficiently understandable? Yes___ No___
   Comment____

5. Does the source program compile satisfactorily (if applicable)? Yes___ No___
   Comment____

6. Does the object program run satisfactorily? Yes___ No___
   Comment____

7. Number of test cases run______. Are any restrictions as to data, size, range, etc. covered adequately in description? Yes___ No___
   Comment____

8. Does the Program Meet the minimal standards of the 1620 Users Group? Yes___ No___
   Comment____

9. Were all necessary parts of the program received? Yes___ No___
   Comment____

10. Please list on the back any suggestions to improve the usefulness of the program. These will be passed onto the author for his consideration.

Please return to:

Mr. Richard L. Pratt
Data Corporation
7500 Old Xenia Pike
Dayton, Ohio 45432

Your Name ________________________

Company ________________________

Address ________________________

User Group Code ________________________

THIS REVIEW FORM IS PART OF THE 1620 USER GROUP ORGANIZATION'S PROGRAM REVIEW AND EVALUATION PROCEDURE. NONMEMBERS ARE CORDIALLY INVITED TO PARTICIPATE IN THIS EVALUATION.

11/09/64

STEVENS INSTITUTE OF TECHNOLOGY

DAVIDSON LABORATORY
CASTLE POINT STATION
HOBOKEN, NEW JERSEY

COMPUTATION OF BESSEL FUNCTIONS OF INTEGRAL ORDER

by

C. E. Grosch

January 1962

Computation of Bessel Functions of Integral Order

by

C. E. Grosch

Many problems involving axi-symmetric potentials have as a final step the evaluation of infinite series of Bessel functions of integral order. That is, the solution may be expressed in terms of an infinite series, each term of which contains a Bessel function of integral order, or the solution is expressible as an integral, the integrand of which contains one or more infinite series of Bessel functions. Since the summing of the series or the numerical quadrature is tedious and time consuming when performed on a desk calculator there is an obvious need for a computer program to generate Bessel functions. This program can then be incorporated in programs to sum the series or do the numerical integration.

Such a program is described in this note. This program is based on an analysis presented in Ref. (1). Since Ref. (1) contains an error, the corrected analysis will be presented below.

Analysis

The formula

$$J_{n+1}(x) = \frac{2n}{x} J_n(x) - J_{n-1}(x) \qquad (1)$$

for generating Bessel functions of the first kind for fixed x, starting from $J_o(x)$ and $J_1(x)$ is satisfactory so long as $n < x$. If $n > x$ the process becomes unstable. If one chooses, however an N sufficiently large compared to x, so that $J^*_{N+1}(x) = 0$, to some order of approximation and $J^*_N(x) << 1$ but not zero then the use of the recursion formula, in a backwards manner is a stable process.

Since $N > x$ and $N >> 1$ one can use as a starting approximation(2)

$$J^*_N(\epsilon N) = \frac{\epsilon^N \exp[N(1-\epsilon^2)^{1/2}]}{\sqrt{2\pi N}\,(1-\epsilon^2)^{1/4}\,[1+\sqrt{1-\epsilon^2}]^N} \qquad (2)$$

$$0 < \epsilon < 1$$

Assume that each $J^*_n(x)$ differs from the correct value $J_n(x)$, i. e.

$$J_n(x) = K_n J^*_n(x) \qquad (3)$$

Since, to our order of approximation

$$J_{N-1}(x) = \frac{2N}{x} J_N(x) \qquad (4)$$

and

$$J^*_{N-1}(x) = \frac{2N}{x} J^*_N(x) \qquad (5)$$

$$K_{N-1} = K_N \qquad (6)$$

In fact, it is easily shown that

$$K_0 = K_1 = \ldots = K_{N-1} = K_N = K \qquad (7)$$

Using the identity(2)

$$1 = J_o(x) + \sum_{n=1}^{\infty} J_{2n}(x) \qquad (8)$$

$$\frac{1}{K} = J^*_o(x) + \sum_{n=1}^{N} J^*_{2n}(x) \qquad (9)$$

It is now only necessary to determine a minimum N large enough so that the process is sufficiently accurate. This has been done in Ref. (1) where the following relations are given

$$10^{-51} \leq x \leq 10 \qquad N - 35.0/(3.5 - \ln x) \qquad (10)$$

$$10 \leq x \leq 500 \qquad N - (21/20)x + 25 \qquad (11)$$

Range

Since the power series expansion of $J_n(x)$ converges rapidly for $x << 1$ and the asymptotic series is very accurate for $x >> 1$, the program was written for the intermediate range where neither the power series nor asymptotic series is useful. The range chosen for this program is $0.001 \leq x \leq 200.0$.

Accuracy

The accuracy of this program has been checked by computing $J_n(x)$ for x = 0.001, 0.100, 1.000, 3.000, 5.000, 7.000, 10.000, 100.000 and 200.000, and comparing with tabulated values. The largest error is ± 1 in the seventh significant digit.

Running Time

Approximately five values of $J_n(x)$ are computed per second. Therefore the running time is approximately 0.2N sec., that is the running time, T, is, in seconds

$$T \approx \begin{cases} 7.0/(3.5-\ln x) & 0.001 \le x \le 10 \\ 5.0 + 0.2x & 10 \le x \le 200 \end{cases}$$

Input - Output

The input, output for this program is the IBM 1622 Card Read-Punch unit.

Likely Changes

The main use of this program is believed to be as part of a larger program which utilizes Bessel Functions in some manner. One would then remove the input-output statements. Another modification could occur if only J_m, J_n, J_j, ... were needed. These can be easily located by noting that

$$\begin{aligned} C(1) &= J_{N+1} \\ C(2) &= J_N \\ &\vdots \\ C(N+2) &= J_o \end{aligned}$$

Then $C(N+2-m) = J_m$ and so on.

Finally, there may be need for the Bessel Functions of the second kind $Y_n(x)$. These are easily found once the $J_n(x)$ are known by using(1)

$$Y_o(x) = (2/\pi)\,[J_o(x)\,(\gamma + \ln \tfrac{x}{2}) - 2 \sum_{n=1}^{N} (-1)^n J_{2n}(x)]$$

$$\gamma = 0.57721566$$

$$Y_1(x) = [J_1(x)Y_o(x) - \tfrac{2}{\pi x}]/J_o(x)$$

and the recurvise relation

$$Y_{n+1}(x) = \frac{2n}{x} Y_n(x) - Y_{n-1}(x)$$

Description of Program

The value of x is read in and punched out. A1 is equal to $(2\pi)^{-1/2}$. The next three statements initialize sum and the C_i. The following four statements, through 4, calculate N in floating point and truncate to a fixed point integer. The formulas for N, (10) and (11) are for a minimum N, therefore 1.0 has been added to ensure that the minimum is exceeded.

In the program

$$C_1 = J_{N+1}(x)$$
$$C_2 = J_N(x)$$
$$C_3 = J_{N-1}(x)$$
$$\vdots$$
$$C_{N+2} = J_0(x)$$

therefore the final value of i is N+2. Statement 5 computes this final value. The next six statements compute C_2 using eq. (2). The statements through 6 compute the C_i, i.e. the $J_n^*(x)$. The four statements following 6 determine whether or not N is odd or even. If N is even then the sum

$$\sum_{i=2,4,6}^{N} C_i = \sum_{n=1}^{N} J_{2n}^*$$

is carried out in statements 7 and 9. The next two statements add to this sum

$$\sum_{i=2,4,6}^{N+2} C_i = \sum_{n=0}^{N} J_{2n}^*$$

If N is odd the corresponding sums (over odd i, i.e. even n) are calculated in statements 8 through 12.

In either case, 13 computes K and the statements through 14, compute n, $J_n(x)$ and punch them. The program then cycles back to read another x.

Sample printouts are appended.

References

(1) Randels, J. B. and Reeves, R.F.: "Note on Empirical Bounds for Generating Bessel Function" Comm. Ass. Comp. Mach., 1, 5, 3, (1958).

(2) Watson, G. N. Theory of Bessel Functions, Cambridge, (1952).



FLOW CHART FOR COMPUTATION OF
BESSEL FUNCTIONS


```
C     COMPUTATION OF BESSEL FUNCTIONS,FIRST KIND,INTEGRAL ORDER
C     FOR ARGUEMENTS IN THE RANGE,GREATER THAN 0.001,LESS THAN 200.0
   99 FORMAT (F13.3)
   98 FORMAT (I9,E16.8)
      DIMENSION C(240)
   14 READ 99,X
      PUNCH 99,X
      A1=0.39894240
      SUM=0.0
      DO 2 I=1,240
    2 C(I)=0.0
      IF (X-10.0) 3,3,4
    3 N=(35.0/(3.5-LOG(X)))+1.0
      GO TO 5
    4 N=1.05*X+26.0
    5 IEND=N+2
      FLN=N
      E=X/FLN
      Y=1.0-E*E
      A=SQRT(Y)
      A2=(E**FLN)*((FLN*A)**(-0.5))*((1.0+A)**(-FLN))
      C(2)=A1*A2*EXP(FLN*A)
      X1=2.0/X
      DO 6 I=1,N
      M=N-I+1
      FLM=M
    6 C(I+2)=FLM*X1*C(I+1)-C(I)
      D=0.5*FLN
      J=D
      FLJ=J
      IF (D-FLJ) 7,7,8
    7 K=2
      GO TO 9
    8 K=1
    9 DO 10 I=K,N,2
   10 SUM=SUM+C(I)
      DO 11 I=K,IEND,2
   11 SUM=SUM+C(I)
      FLK=1.0/SUM
      DO 12 I=1,IEND
   12 C(I)=FLK*C(I)
      DO 13 I=1,IEND
      NP=N-I+2
   13 PUNCH 98,NP,C(I)
      GO TO 14
      END
```

SAMPLE PROGRAM

X
1.000

n	$J_n(X)$
12	.00000000E-99
11	.11957287E-10
10	.26306032E-09
9	.52492492E-08
8	.94223427E-07
7	.15023257E-05
6	.20938336E-04
5	.24975771E-03
4	.24766388E-02
3	.19563353E-01
2	.11490349E-00
1	.44005062E-00
-0	.76519775E-00

X
10.000

n	$J_n(X)$
31	.00000000E-99
30	.15069845E-11
29	.90419072E-11
28	.50936078E-10
27	.27620013E-09
26	.14405447E-08
25	.72146326E-08
24	.34632619E-07
23	.15902195E-06
22	.69686837E-06
21	.29071990E-05
20	.11513368E-04
19	.43146273E-04
18	.15244247E-03
17	.50564662E-03
16	.15667560E-02
15	.45079727E-02
14	.11957163E-01
13	.28972083E-01
12	.63370254E-01
11	.12311653E-00
10	.20748612E-00
9	.29185572E-00
8	.31785417E-00
7	.21671095E-00
6	-.14458835E-01
5	-.23406155E-00
4	-.21960271E-00
3	.58379378E-01
2	.25463033E-00
1	.43472757E-01
-0	-.24593579E-00